[  Please note:  this Codebook does not include the code in Codebook 1 & Codebook 2. You can download  Codebook 1 & Codebook 2 from the website. 

                     Code Book 3

                      I hope all these code books are helping everone out on there games!!! As for the code books I will continue to make newer and better versions. I really enjoy making these books for everyone to use and to help on there games.

 NOTE:  If you have any codes that are not in here send them to me so that they may be added to later versions, my email is   cloud65559@yahoo.com         Thankyou ! ! !

Helpers:

Flamemage 999(alot of the codes you see)

brently_11 (gave me the nameing code, THANKS!!!!!)

rpgdude(gave "invincability code)

vegarelde(snapshot print code)











True Predator

>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>

Legend: 

Scripts = all scripts will appear in red

Code = all codes will appear in green

Doc. = descriptions and such will stay black and appears next to the script or code

edit = where you edit or replace with your  own object or text

Event =In game events(Creation ,Step, Drawing, etc..)

>>>>> = this is a separator between codes and scripts

>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>

"Load Background"

This code like it says loads backgrounds. Good for people making programs that have customizable "skins" like some of the media players that have been made(about 100 of them!). 

{

fname=get_open_filename('BMP files(*.BMP)*.BMP','')

background_replace(fname,DBG); 

}

>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>

"Shooter Movement"

Basic movement but adds the border to it.

////////////////////////////////////////////////////////////////

// script name: Shooter Movement

//     creator: Chris Johns (FlameMage99@aol.com)

//        date: September 23, 2001

//     updated: N/A

//

// description: This is used to move an object while inside the

//              room borders, and stop it when no key is pressed.

//              It also uses the space key to create a bullet.

//

//   arguments: N/A

//

//     remarks: Call from / place in the step event.

//              Define canshoot as 1 in the creation and alarm[1]

//              events.

//

//////////////////////////////////////////////////////////////////

{

 // Movement, stopping when a boundary is hit.

 if (keyboard_check(vk_up))

  {

   direction = 90;

   if (y > 0) speed = 12; else speed = 0;

  }

 if (keyboard_check(vk_down))

  {

   direction = 270;

   if (y < room_height - sprite_height) speed = 12; else speed = 0;

  }

 if (keyboard_check(vk_left))

  {

   direction = 180;

   if (x > 0) speed = 12; else speed = 0;

  }

 if (keyboard_check(vk_right))

  {

   direction = 0;

   if (x < room_width - sprite_width) speed = 12; else speed = 0;

  }

 // Stopping when no arrows are pressed

 if not(keyboard_check(vk_up) || keyboard_check(vk_down) || keyboard_check(vk_left) || keyboard_check(vk_right))

  {

   speed = 0;

  }

 // Shooting Bullets

 if (keyboard_check(vk_space) && canshoot)

  {

   instance_create(x-4+sprite_width/2,y+4,bullet); canshoot = 0;

   alarm[1] = 10; // set the delay (in steps)

  }

}

>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>

“Random Background Music”

Very Very good, I like this and had been waiting for something like this code, well done!

////////////////////////////////////////////////////////////////

// script name: Random Music

//     creator: Chris Johns (FlameMage99@aol.com)

//        date: August 31, 2001

//     updated: N/A

//

// description: Selects a random sound to loop from an array.

//              To add or remove more sounds, the random(4)

//              should be changed to the last number in the array.

//              Don't forget to change the sounds to ones that

//              are in your game!

//

//   arguments: N/A

//

//     remarks: Call from the creation event.

//

//////////////////////////////////////////////////////////////////

{

 // bgsound array

 bg[0] = BGmusic1;

 bg[1] = BGmusic2;

 bg[2] = BGmusic3;

 bg[3] = BGmusic4;

 bg[4] = BGmusic5;

 // Choose random bgmusic

 sound_loop(bg[round(random(4))]);

}

>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>

“Damage and Explode”

Simplifies the basis of changing the sprite with damage

////////////////////////////////////////////////////////////////

// script name: Damage and Explode

//     creator: Chris Johns (FlameMage99@aol.com)

//        date: August 25, 2001

//     updated: September 4, 2001

//

// description: Used to index a sprite depending on damage,

//              then switch to a sprite named 'explosion'

//

//   arguments: N/A

//

//     remarks: Define global.damage in the creation as 0. Call

//              this from the step event.

//

//////////////////////////////////////////////////////////////////

{ 

 if (global.Damage < 4) // 4 is the number of hits to take.

  {

   image_single = global.Damage;

  }

 else

  {

   if (sprite_index != explosion)

    {

     sprite_index = explosion;

     image_single = -1;

    }

  }

 if (sprite_index == explosion && image_index == image_number - 1) instance_destroy();

}

>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>

“Border View Destroy”

Instead of the object being destroyed outside the screen its done in side, NICE!!!

////////////////////////////////////////////////////////////////

// script name: View Border

//     creator: Chris Johns (FlameMage99@aol.com)

//        date: September 15, 2001

//     updated: N/A

//

// description: This is used to destroy an instance when it

//              reaches the border of the view, rather than the

//              border of the room. You can replace

//              'instance_destroy()' with which ever action you

//              want to use.

//

//   arguments: N/A

//

//     remarks: Call from the step event.

//

//////////////////////////////////////////////////////////////////

{

 if (x <= view_left[0] || y <= view_top[0] ||

 x >= view_left[0] + view_width[0] ||

 y >= view_top[0] + view_height[0])

  {

   instance_destroy();

  }

}

>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>

“Open .WAV, .avi, .Wmp, and Play”

Yes, this is a very good code, I’d recommend using it a lot. 

////////////////////////////////////////////////////////////////

// script name: Open file and play

// creator: Vegar Elde (vegarelde@hotmail.com)

// date: November 9 2001

//

//

// description: This script is used to first open a video file

// and then play it internally.You can even

// play Wave mp3 Wma

// and video files if you change the .avi

//

// arguments: N/A

//

// remarks:Now you can make a Video and audio player!!!

//

//////////////////////////////////////////////////////////////////

{

fname = get_open_filename('file(*.avi)|*.avi','')

show_video(fname,false,0);

}

>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>

“Enemy Respawn”

Very nice piece of code(he’s on a role), Excellent for side scrollers and maze games.

////////////////////////////////////////////////////////////////

// script name: Enemy Respawn

//     creator: Chris Johns (FlameMage99@aol.com)

//        date: September 23, 2001

//     updated: N/A

//

// description: This will respawn the enemy specified for the

//              room 5 times, then it will change rooms. To

//              define the enemies for each room, you need to

//              add the array with the others like so:

//              EnemyType['room number'] = 'enemy to use';

//              The first room starts at 0.

//

//   arguments: N/A

//

//     remarks: Call from the step event.

//              You need to define global.DeadEnemy and

//              global.EnemyWave as 0 in the creation.

//              Increase global.DeadEnemy relative to 1 in the

//              destroy event of all enemies.

//

//////////////////////////////////////////////////////////////////

{

 // Define the mosters for each room

 EnemyType[0] = batcat; // first room

 EnemyType[1] = night_fairy; // second room

 // Respawn or Change room

 if (global.DeadEnemy == 5)

  {

   global.DeadEnemy = 0;

   if (global.EnemyWave < 4) global.EnemyWave += 1; else room_goto_next();

   repeat (5) instance_create(room_width/2-(100-(round(random(200)))),room_height/2-(100-(round(random(200)))),EnemyType[room]);

  }

}

>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>

“Game Save Prompt”

I like this code, very professional!!! 

////////////////////////////////////////////////////////////////

// script name: Game Save Prompt

//     creator: Chris Johns (FlameMage99@aol.com)

//        date: September 9, 2001

//     updated: N/A

//

// description: This is used to create a saved game file. It'll

//              ask to replace the file if it already exists.

//

//   arguments: N/A

//

//     remarks: The file extension should be provided by the user.

//              Call from any event to trigger a save prompt, for

//              example, a key or mouse button.

//

//////////////////////////////////////////////////////////////////

{

 SaveAs = get_save_filename('Saved Games *.sav|*.sav','Untitled.sav');

 mouse_clear(mb_left);

 if (file_exists(SaveAs))

  {

   if not(show_question('Warning!'+chr(13)+SaveAs+' Already exists!'+chr(13)+'Do you want to overwrite it?'))

    {

     mouse_clear(mb_left);

     exit;

    }

  }

 game_save(SaveAs);

}

>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>

"Move_Aligned"

Ver Very nice peice of code!!!

////////////////////////////////////////////////////////////////

// script name: Move_Aligned

//     creator: Chris Johns (FlameMage99@aol.com)

//        date: September 2, 2001

//     updated: N/A

//

// description: Checks for colisions before moving in the

//              specified direction. It will also ignore keyboard

//              input unless the player is 'snapped to a grid' at

//              16 x 16 which should make crossing narrow paths

//              easier. The snapping can cause problems if the

//              room is not the appropriate dimensions. The player

//              will stop movement only when the arrow keys are

//              not pressed. Unlike the no key event, which can

//              create bugs when used for that purpose.

//             

//   arguments: N/A

//

//     remarks: Place in the step event.

//              This was originally done for Chronic's RPG Engine.

//              Good luck Chronic!

//

//////////////////////////////////////////////////////////////////

{

 if (place_snapped(16,16))

  {

   if (keyboard_check(vk_left))

    {

     if (place_free(x-4,y)) speed = 4 else speed = 0;

     direction = 180;

     image_single = -1;

     sprite_index = player_left;

    }

   if (keyboard_check(vk_up))

    {

     if (place_free(x,y-4)) speed = 4 else speed = 0;

     direction = 90;

     image_single = -1;

     sprite_index = player_up;

    }

   if (keyboard_check(vk_right))

    {

     if (place_free(x+4,y)) speed = 4 else speed = 0;

     direction = 0;

     image_single = -1;

     sprite_index = player_right;

    }

   if (keyboard_check(vk_down))

    {

 if (place_free(x,y+4)) speed = 4 else speed = 0;

     direction = 270;

     image_single = -1;

     sprite_index = player_down;

    }

   if not(keyboard_check(vk_left) || keyboard_check(vk_up) ||

   keyboard_check(vk_right) || keyboard_check(vk_down))

    {

     speed = 0;

     image_single = image_index;

    }

  }

}

>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>

"MP3 file execute"

Finally able to use mp3's without so many problems, very nice code.

////////////////////////////////////////////////////////////////

// script name: File Execute

//     creator: Chris Johns (FlameMage99@aol.com)

//        date: September 5, 2001

//     updated: N/A

//

// description: This opens a file if it can be found. The mask

//              is for mp3 files only, and uses execute_shell,

//              so will not work in secure mode.

//

//   arguments: N/A

//

//     remarks: Call from the step event

//

//////////////////////////////////////////////////////////////////

{

 if (mouse_check_button(mb_right))

  {

   bgmusic = get_open_filename('Mpeg Layer 3 (*.mp3)|*.mp3','mp3');

   mouse_clear(mb_right);

   if (file_exists(bgmusic)) execute_shell(bgmusic,0);

  }

}

>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>

